
Данная статья продолжает тему об использовании фильтров с кварцевым и 

керамическим резонаторами для преобразования меандра в синусоидальный 

сигнал. В статье рассматривается способ определения ширины полосы 

пропускания фильтра при помощи перестраиваемого в узком диапазоне частот 

генератора, построенного на основе микросхемы AD9833 и контроллера Arduino. 

Полоса пропускания – это диапазон частот, в пределах которого 

амплитудно-частотная характеристика (АЧХ) электронного устройства 

достаточно равномерна для того, чтобы обеспечить передачу сигнала без 

существенного искажения его формы. 

Ширина полосы обычно определяется как разность верхней и нижней 

граничных частот участка АЧХ, на котором амплитуда колебаний равняется 

единице, деленной на квадратный корень из двух от максимальной (приближенно 

0,7). 

Проблема с определением ширины полосы пропускания фильтров, 

содержащих в своем составе кварцевые резонаторы, заключается в том, что у 

подобных фильтров полоса пропускания очень узкая: она может составлять одну 

десятитысячную долю от рабочей частоты фильтра. Далеко не всякий 

лабораторный генератор сигналов может обеспечить необходимую для решения 

данной задачи плавность перестройки частоты. 

Микросхема AD9833 представляет собой самый простой из синтезаторов 

сигналов, выпускаемых фирмой Analog Devices. Она способна формировать 

сигналы синусоидальной, треугольной и прямоугольной формы с частотой от 0 до 

12,5 МГц (при использовании задающего тактового генератора с частотой 

25 МГц). 

Частота и фаза выходного сигнала AD9833 являются программируемыми. 

Программирование AD9833 осуществляется с внешнего микроконтроллера по 

интерфейсу SPI, причем можно использовать контроллеры различных типов, в 

том числе – простые в эксплуатации и широко распространенные контроллеры 

линейки Arduino. 

Контроллер Arduino позволяет выполнять ввод информации с помощью 

самых разнообразных технических средств: можно, например, вводить данные с 

помощью клавиатуры или энкодера, а также передавать данные с компьютера или 

ноутбука по интерфейсу USB. 

На рисунке 1 в качестве примера показана принципиальная схема 

генератора синусоидального сигнала, в котором используется самый простой 

способ ввода данных: при помощи потенциометра R1. Напряжение, подаваемое с 

потенциометра на аналоговый вход контроллера A0, будет преобразовано 

аналого-цифровым преобразователем (АЦП) в целое число, находящееся в 

диапазоне от 0 до 1023. 

 

 

 
 

Рисунок 1. Принципиальная схема генератора 

 

Тактовый сигнал с частотой 25 МГц подается на синтезатор DA1 c 

кварцевого генератора DD1. 

Код К выходной частоты F для синтезатора вычисляется по следующей 

формуле: 

 

K = F × 228 / 25000000. 

 

Результат вычислений должен быть округлен до 32-разрядного целого числа 

без знака, а затем разделен на две части, содержащие по 14 разрядов каждая 

(старшие 4 разряда полученного кода содержат нули и просто отбрасываются). 

Обе части кода поочередно должны быть загружены в регистр синтезатора, 

хранящий код частоты. 

Напряжение питания +5В, необходимое для работы кварцевого генератора 

DD1 и синтезатора синусоидального сигнала DA1, обеспечивает стабилизатор 

напряжения, встроенный в контроллер Arduino. 

Выходной сигнал синтезатора AD9833 на высоких частотах имеет заметную 

ступенчатость, поэтому вначале его приходится пропускать через RC-цепочку, 

представляющую собой фильтр нижних частот, и только потом подавать на вход 

усилителя, построенного на микросхеме DA2 типа AD8055. С выхода усилителя 



сигнал поступает на выход генератора, к которому подключается тестируемый 

фильтр на кварцевых или керамических резонаторах. 

Контроллер обеспечивает перестройку частоты сигнала синтезатора в узком 

диапазоне, ширина которого определяется загруженной в него программой. 

Программа для контроллера Arduino (скетч) приведена в листинге 1. 

В данном примере диапазон перестройки частоты задан равным 3,2 кГц: от 

1,9984 МГц до 2,0016 МГц, Шаг приращения частоты сигнала при повороте ручки 

потенциометра задан равным 25 Гц. 

 

 

Листинг 1. Программа для управления работой синтезатора AD9833 
#include <SPI.h> 

#define pot A0  // сигнал с потенциометра подается на вход A0 

 

const float fgen = 2.5E7;  // частота сигнала ГТИ, Гц 

const float sf = 1.9984E6; // нижнее значение выходной частоты, Гц 

const float df = 2.5E1;    // шаг приращения частоты, Гц 

 

unsigned int a0old; // предыдущее значение сигнала от потенциометра 

unsigned int a0new; // новое значение сигнала от потенциометра 

float ff;           // код частоты в формате с плавающей точкой 

unsigned long f;    // двоичный код выходной частоты 

unsigned long lf;   // младшая часть значения выходной частоты 

unsigned long hf;   // старшая часть значения выходной частоты 

unsigned int flsb;  // младшее слово кода частоты с кодом регистра 

unsigned int fmsb;  // старшее слово кода частоты с кодом регистра 

 

void setup() { 

  // Объявляем A0 как входной пин 

  pinMode(pot, INPUT);  

  // Обнуляем переменную, содержащую предыдущее 

  // значение сигнала от потенциометра 

  a0old = 0; 

  // Вычисляем код частоты в формате с плавающей точкой 

  ff = sf*((float)0x10000000l/fgen); 

  // Получаем двоичный код выходной частоты и делим его на 

  // младшую и старшую две части 

  f = (unsigned long)ff; 

  lf = (f & 0x3fffl) | 0x4000l; 

  flsb = (unsigned int)lf; 

  hf = (f >> 14) | 0x4000l; 

  fmsb = (unsigned int)hf; 

  // Запускаем интерфейс SPI 

  SPI.begin(); 

  // Сбрасываем генератор в исходное состояние 

  WriteAD9833(0x2100); 

  // Загружаем код частоты 

  WriteAD9833(flsb); 

  WriteAD9833(fmsb); 

  // Устанавливаем сдвиг по фазе равным нулю 

  WriteAD9833(0xC000); 



  // Запускаем генератор 

  WriteAD9833(0x2000); 

} 

 

// Процедура для передачи данных синтезатору AD9833 

// через интерфейс SPI 

void WriteAD9833(uint16_t Data){ 

  SPI.beginTransaction(SPISettings(SPI_CLOCK_DIV2, MSBFIRST, 

SPI_MODE2)); 

  digitalWrite(SS, LOW); 

  delayMicroseconds(1); 

  SPI.transfer16(Data); 

  digitalWrite(SS, HIGH); 

  SPI.endTransaction(); 

} 

 

// Процедура, реализующая микроконтроллерный цикл обработки данных 

void loop() { 

  // Считываем сигнал с потенциометра и загрубляем результат 

  // путем деления полученного числового значения на 8 

  a0new = ((unsigned int)analogRead(pot))>>3; 

  // Если новое значение сигнала потенциометра отличается 

  // от старого, то вычисляем и перезагружаем в синтезатор 

  // новый код частоты 

  if(a0new != a0old) 

  { 

    // Копируем новое значение сигнала потенциометра в 

    // переменную, хранящую предыдущее значение 

    a0old = a0new; 

    // Вычисляем код частоты в формате с плавающей точкой 

    ff = ((sf + df*(float)a0old)*(float)0x10000000l)/fgen; 

    // Получаем двоичный код выходной частоты и делим его на 

    // младшую и старшую две части 

    f = (unsigned long)ff; 

    lf = (f & 0x3fffl) | 0x4000l; 

    flsb = (unsigned int)lf; 

    hf = (f >> 14) | 0x4000l; 

    fmsb = (unsigned int)hf; 

    // Загружаем новый код частоты в синтезатор 

    WriteAD9833(0x2000); 

    WriteAD9833(flsb); 

    WriteAD9833(fmsb); 

  } 

  // Задержка на 200 мс 

  delay(200); 

 } 

 

 

Контроллер производит проверку сигнала от потенциометра 5 раз в секунду. 

При изменении значения этого сигнала он загружает в синтезатор AD9833 новый 

код частоты. 

